

Textul si imaginile din acest document sunt licentiate

Attribution-NonCommercial-NoDerivs
CC BY-NC-ND



Codul sursa din acest document este licentiat


Public-Domain

Esti liber sa distribui acest document prin orice mijloace consideri (email, publicare pe website / blog, printare, sau orice alt mijloc), atat timp cat nu aduci nici un fel de modificari acestuia. Codul sursa din acest document poate fi utilizat in orice fel de scop, de natura comerciala sau nu, fara nici un fel de limitari.

Arduino – RTC

Daca doresti ca placa Arduino sa interactionezeze cu notiunea de timp, atunci iti este necesar un RTC sau Real Time Clock. RTC-ul este un circuit integrat creat special pentru a mentine timpul, adica functioneaza exact ca un ceas. In acest mod, placa Arduino nu este obligata sa mentina singura timpul, deoarece nu este destinata pentru acest lucru.

In majoritatea situatiilor nu este necesar ca un microcontroller sa cunoasca timpul, dar cum procedezi atunci cand vrei sa construiesti o alarma, un ceas, cronometru sau un data logger ? In acest caz iti este necesar un RTC DS1307Z+, ceas foarte popular care functioneaza la o tensiune de 5V impreuna cu placa Arduino.



Pentru o aplicatie simpla, in care placa Arduino trebuie sa afiseze timpul vei avea nevoie de urmatoarele componente:



- O placa Arduino - <http://www.robofun.ro/arduino>
- Un Real Time Clock DS1307Z+

http://www.robofun.ro/index.php?route=product/product&product_id=1333

- Fire de conexiune - <http://www.robofun.ro/cabluri>
- O bareta de pini - http://www.robofun.ro/conectori/bareta_40pini
- O baterie litiu 3V CR1216 - <http://www.robofun.ro/baterie-CR1216>


Cum se conecteaza placuta RTC cu placa Arduino ?


Înainte de a conecta RTC-ul împreună cu placa Arduino, trebuie să realizezi câteva lipituri simple cu leașonul asupra plăcuței RTC. Tot ce trebuie să faci este să lipești bareta de pini în gaurile plăcuței RTC (doar 6 pini îți sunt necesari). Vezi cum s-au realizat lipiturile în imaginile de mai jos:



Dupa ce ai realizat lipiturile urmeaza sa conectezi RTC-ul impreuna cu placa Arduino, dupa urmatorul tabel. Te vei folosi de firele de conexiune cu capete mama-tata si tot ce trebuie sa faci este sa infigi firele in pinii placutei RTC si in pinii placii Arduino:

Arduino pin 5V	RTC DS1307Z+ pin 5V
Arduino pin GND	RTC DS1307Z+ pin GND
Arduino pin SDA	RTC DS1307Z+ pin SDA
Arduino pin SCL	RTC DS1307Z+ pin SCL





Urmeaza sa conectezi bateria ceasului care este absolut necesara, altfel RTC-ul nu ar functiona corect. Bateria se conecteaza cu borna + spre exterior, asa cum iti indica si slotul bateriei.

Sketch-ul Arduino ?

Tot ce iti ramane de facut, este sa incarci in placa Arduino un sketch care iti permite sa vezi care este timpul si totodata sa il reglezi. Copiaza codul sursa de mai jos, incarca-l in Arduino si imediat dupa ce ai facut acest lucru deschide Monitorul Serial. Vei observa ca la fiecare secunda iti vor aparea mai multe informatii: data si ora.

```
#include "Wire.h"
#define DS1307_ADDRESS 0x68
byte zero = 0x00; //workaround for issue #527

void setup(){
    Wire.begin();
    Serial.begin(9600);
    setDateTime(); //MUST CONFIGURE IN FUNCTION
}

void loop(){
    printDate();
    delay(1000);
}

void setDateTime(){

    byte second =      45; //0-59
    byte minute =      17; //0-59
    byte hour =        22; //0-23
    byte weekDay =      2; //1-7
    byte monthDay =     16; //1-31
    byte month =        5; //1-12
    byte year =        14; //0-99

    Wire.beginTransmission(DS1307_ADDRESS);
    Wire.write(zero); //stop Oscillator

    Wire.write(decToBcd(second));
    Wire.write(decToBcd(minute));
    Wire.write(decToBcd(hour));
    Wire.write(decToBcd(weekDay));
    Wire.write(decToBcd(monthDay));
    Wire.write(decToBcd(month));
```

```

    Wire.write(decToBcd(year));
    Wire.write(zero); //start
    Wire.endTransmission();

}

byte decToBcd(byte val){
// Convert normal decimal numbers to binary coded decimal
    return ( (val/10*16) + (val%10) );
}

byte bcdToDec(byte val) {
// Convert binary coded decimal to normal decimal numbers
    return ( (val/16*10) + (val%16) );
}

void printDate(){

    // Reset the register pointer
    Wire.beginTransmission(DS1307_ADDRESS);
    Wire.write(zero);
    Wire.endTransmission();

    Wire.requestFrom(DS1307_ADDRESS, 7);


    int second = bcdToDec(Wire.read());
    int minute = bcdToDec(Wire.read());
    int hour = bcdToDec(Wire.read() & 0b111111); //24 hour time
    int weekDay = bcdToDec(Wire.read()); //0-6 -> sunday - Saturday
    int monthDay = bcdToDec(Wire.read());
    int month = bcdToDec(Wire.read());
    int year = bcdToDec(Wire.read());

    //print the date EG   3/1/11 23:59:59
    Serial.print(month);
    Serial.print("/");
    Serial.print(monthDay);
    Serial.print("/");
    Serial.print(year);
    Serial.print(" ");
    Serial.print(hour);
    Serial.print(":");
    Serial.print(minute);
    Serial.print(":");
    Serial.println(second);

}

```

Daca doresti sa reglezi ceasul trebuie mai intai sa modifichi variabilele second, minute, hour, weekDay, monthDay, month, year din functia setTime(). Imediat cum ai modificat aceste variabile trebuie sa incarci din nou sketch-ul in placa Arduino si sa deschizi din nou Monitorul Serial.



Daca doresti sa afli cum se poate construi un ceas care sa iti arate timpul printr-un afisaj LCD, atunci acceseaza link-ul de mai jos:

<http://www.electroschematics.com/8921/digital-clock-with-arduino-and-ds1307/>

Daca doresti sa afli cum se poate construi un data logger care sa inregistreze date la intervale precise de timp folosind un RTC, atunci acceseaza link-urile de mai jos:

<http://www.bajdi.com/arduino-temperature-logging-and-webserver-with-rtc/>

<http://jobs.arduinoexperts.com/2013/03/02/simple-data-logging-using-arduino/>

<http://www.robofun.ro/forum>